# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=angr-py2
_pkgname=angr
pkgver=7.8.9.26
pkgrel=5
epoch=1
groups=('blackarch' 'blackarch-binary' 'blackarch-disassembler'
        'blackarch-reversing')
pkgdesc='The next-generation binary analysis platform from UC Santa Barbaras Seclab.'
arch=('x86_64' 'aarch64')
url='https://pypi.org/project/angr/#files'
license=('BSD')
depends=('python2' 'python2-capstone' 'python2-networkx' 'python2-rpyc'
         'python2-progressbar' 'python2-mulpyplexer' 'python2-unicorn'
         'python2-cooldict' 'python2-archinfo' 'python2-pyvex' 'python2-ana'
         'python2-claripy' 'python2-simuvex' 'python2-cle' 'libffi'
         'python2-cachetools' 'python2-pyelftools' 'python2-cffi'
         'python2-bintrees' 'python2-dpkt' 'python2-z3' 'python2-gitpython'
         'python2-sortedcontainers' 'python2-pycparser' 'python2-ailment')
makedepends=('git' 'python2-setuptools')
source=("https://files.pythonhosted.org/packages/source/${_pkgname::1}/$_pkgname/angr-$pkgver.tar.gz")
sha512sums=('ef90d12ff1029fdd20ba884ee95fb7ce97bad6b9cfd530e983922b36e396b0c587728f98fc2bec1e7422f272b5e27a3df12c282038b01b157230ba265b1c1c1e')

build() {
  cd "$_pkgname-$pkgver"

  python2 setup.py build
}

package() {
  cd "$_pkgname-$pkgver"

  python2 setup.py install --root="$pkgdir" --prefix=/usr -O1 --skip-build
}

